
*------------------------------------------------------------------------------------
* Table A3 -- Heterogeneity in Hosting Support Impacts
*------------------------------------------------------------------------------------

local num = 0
foreach var in h_female refugee_facilitator_ever_flag h_profit h_domain1 h_domain4 h_domain6 h_domain10 h_domain7 h_domain8 h_domain3 h_mentor_profit {
	local ++num
	gen X = `var' // to produce table
	
	if !inlist("`var'", "h_mentor_profit", "refugee_facilitator_ever_flag", "h_domain3") {
		pdslasso e_domain1 ib6.treatment##ib0.X (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
		eststo h12_`num'
	}	
	
	if inlist("`var'", "h_mentor_profit") { // Omit the uninteracted control and irrelevant interactions
		pdslasso e_domain1 ib6.treatment 1.treatment#ib0.X 2.treatment#ib0.X (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
		eststo h12_`num'
	}
	
	if inlist("`var'", "refugee_facilitator_ever_flag") { // Omit the uninteracted control and irrelevant interactions
		pdslasso e_domain1 ib6.treatment 5.treatment#ib0.X 4.treatment#ib0.X (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
		eststo h12_`num'
	}
	
	if inlist("`var'", "h_domain3") {
	local exclude cat_b_f15_*
	local temp_cat_list : list global(cat_list) - exclude // don't include dimension of heterogeneity in lasso controls
	
		pdslasso e_domain1 ib6.treatment##ib0.X (i.strata i.wave phone_survey survey_date b_domain1 `temp_cat_list' $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
		eststo h12_`num'
	}	
	
	drop X		
}

esttab h12_* using "$path/Output/Appendix_A/hetero_domain1.tex", label $nostar collabels(none) replace nolines nonumber substitute(\_ _ \$ $) ///
cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N, fmt(%9.0fc) labels("Observations")) ///
keep(1.treatment 2.treatment 3.treatment 4.treatment 5.treatment 1.X  1.treatment#1.X 2.treatment#1.X 3.treatment#1.X 4.treatment#1.X 5.treatment#1.X) ///
order(5.treatment#1.X 5.treatment 4.treatment#1.X 4.treatment 3.treatment#1.X 3.treatment 2.treatment#1.X 2.treatment 1.treatment#1.X 1.treatment 1.X) ///
coeflabels(1.treatment "Mentored by Ugandan" 2.treatment "Mentored by Refugee" 3.treatment "Grant Only" 4.treatment "Information Only" 5.treatment "Info. + Labeled Grant" 1.X "\textit{X}" 1.treatment#1.X "Mentored by Ugandan $\times$ \textit{X}" 2.treatment#1.X "Mentored by Refugee $\times$ \textit{X}" 3.treatment#1.X "Grant Only $\times$ \textit{X}" 4.treatment#1.X "Information Only $\times$ \textit{X}" 5.treatment#1.X "Info. + Labeled Grant $\times$ \textit{X}") ///
mtitle("\shortstack{Female\\Owner}" "\shortstack{Refugee\\Facilitator}" "\shortstack{Business\\Profit}" "\shortstack{Supports\\Hosting\\Index}" "\shortstack{Economic\\Beliefs\\Index}" "\shortstack{Cultural\\Attitudes\\Index}" "\shortstack{Household\\Well-Being\\Index}" "\shortstack{Contact\\Refugees\\(Choice)}" "\shortstack{Contact\\Refugees (Cir-\\cumstance)}" "\shortstack{Knows\\About\\Aid-Sharing}" "\shortstack{Mentor\\Profit}") ///
$stars_setup ///
prehead("\begin{table}[H]	\centering	\scriptsize	\caption{Heterogeneity in Impacts on Support for Refugee Integration Policies} \label{tab:hetero_domain1}	\begin{tabular}{l*{11}{>{\centering\arraybackslash}p{1.35cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-12}") ///
prefoot("& & & & & & & & & & & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{12}{p{\linewidth}}{\footnotesize The dependent variable for each column is the integration policies summary index. Each column title lists the dimension of baseline heterogeneity (\textit{X}) that is analyzed in the regression. \textit{X} denotes above median values for continuous variables. An observation is a surveyed respondent per post-baseline survey round in Uganda. Results estimated through ANCOVA regression with baseline controls selected through double lasso. Standard errors clustered at the enterprise level in parentheses; two-sided $ p $-values in brackets. $stars_note}  \end{tabular} \\ \end{table}%")

estimates drop h12_*


*------------------------------------------------------------------------------------
* Table A4 - Strongly Support and Oppose Policies
*------------------------------------------------------------------------------------

foreach var in d1_1_strongagree d1_1_strongdisagree d1_3_strongagree d1_3_strongdisagree d1_5_strongagree d1_5_strongdisagree d1_7_strongagree d1_7_strongdisagree {
	pdslasso `var' ib6.treatment (i.strata i.wave phone_survey survey_date b`var' $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b`var') post(pds) robust cluster(ent_id) lopt(prestd)
	
		test 4.treatment = 5.treatment
		estadd scalar p1 = r(p)
		test 3.treatment = 5.treatment
		estadd scalar p2 = r(p)
		test 2.treatment = 5.treatment
		estadd scalar p3 = r(p)
		test 2.treatment = 4.treatment
		estadd scalar p4 = r(p)
		test 1.treatment = 2.treatment
		estadd scalar p5 = r(p)
		
		qui sum `var' if treatment == 6 & e(sample)
		estadd scalar post_mean = r(mean)
		qui sum b`var' if treatment == 6 & finish_flag == 1
		estadd scalar b_mean = r(mean)
		
		forvalues i=1/5{
			test `i'.treatment = 0
			local p`i' = r(p)
		}
		mat q = (`p1', `p2', `p3', `p4', `p5')
		mat colnames q = 1.treatment 2.treatment 3.treatment 4.treatment 5.treatment	
		estadd matrix q
			
estimates store est_`var'
}

esttab est_d1_1_strongagree est_d1_1_strongdisagree est_d1_3_strongagree est_d1_3_strongdisagree est_d1_5_strongagree est_d1_5_strongdisagree est_d1_7_strongagree est_d1_7_strongdisagree  using "$path/Output/Appendix_A/strongagree.tex", label collabels(none) replace nolines nonumber keep(1.treatment 2.treatment 3.treatment 4.treatment 5.treatment) cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N b_mean post_mean p1 p2 p3 p4 p5, fmt(%9.0fc %12.2fc %12.2fc %9.2f %9.2f %9.2f %9.2f %9.2f) labels("Observations" "Control Mean: Baseline" "Control Mean: Follow-Ups" "Labeled Grant = Info Only" "Labeled Grant = Grant Only" "Labeled Grant = R-Mentee" "R-Mentee = Info Only" "R-Mentee = U-Mentee")) coeflabels(1.treatment "Mentored by Ugandan" 2.treatment "Mentored by Refugee" 3.treatment "Grant Only" 4.treatment "Information Only" 5.treatment "Info. + Labeled Grant") order(5.treatment 4.treatment 3.treatment 2.treatment 1.treatment) substitute(\_ _ \$ $) ///
$stars_setup ///
prehead("\begin{table}[H]	\centering	\footnotesize	\caption{Strongly Support and Strongly Oppose Inclusive Policies} \label{tab:strongly}	\begin{tabular}{l*{8}{>{\centering\arraybackslash}p{1.85cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-9}") ///
prefoot("& & & & & & & &\\") ///
postfoot("\bottomrule \bottomrule \multicolumn{9}{p{\linewidth}}{\footnotesize An observation is a surveyed respondent per post-baseline survey round in Uganda. Results estimated through ANCOVA regression with baseline controls selected through double lasso. Standard errors clustered at the enterprise level in parentheses. Brackets and the last five rows display two-sided $ p $-values. Outcomes that were not pre-specified are denoted with $^+$. $stars_note}  \end{tabular} \\ \end{table}%")

*------------------------------------------------------------------------------------
* Table A5 -- Policy Preferences with Entropy Balancing
*------------------------------------------------------------------------------------

label var d1_1 "\shortstack{Supports\\Refugee\\Hosting}"
label var d1_3 "\shortstack{Supports\\More\\Refugees}"
label var d1_7 "\shortstack{Supports\\Right\\to Work}"
label var d1_5 "\shortstack{Supports\\Freedom of\\Movement}"
label var e_domain1 "\shortstack{Integration\\Policies\\Index}"
label var oyoh_yes "\shortstack{Supported\\Phone\\Campaign$^+$}"

gen secondary = b_p12 >= 14 if b_p12 != .
gen age_18_30 = inrange(age,18,30) if age != .

local mean_secondary = 0.165 // https://www.ubos.org/wp-content/uploads/publications/09_2021Uganda-National-Survey-Report-2019-2020.pdf Table 3.8 completed secondary + post-secondary and above
local mean_age_18_30 = 0.415 // Table 2.4 of above link, population share 18-30 among those 18+ (0.191 / 0.46)
local mean_support_work = 0.72 // https://www.rescue.org/sites/default/files/document/2858/ircuganda.pdf Fig 8 supports allowing refugees work

ebalfit secondary age_18_30 bd1_7, population(4.5e7: `mean_secondary' `mean_age_18_30' `mean_support_work') baltab generate(entropy_weight)

pdslasso e_domain1 ib6.treatment (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list) [pweight=entropy_weight], partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
		test 4.treatment = 5.treatment
		estadd scalar p1 = r(p)
		test 3.treatment = 5.treatment
		estadd scalar p2 = r(p)
		test 2.treatment = 5.treatment
		estadd scalar p3 = r(p)
		test 2.treatment = 4.treatment
		estadd scalar p4 = r(p)
		test 1.treatment = 2.treatment
		estadd scalar p5 = r(p)
		
		qui sum b_domain1 if treatment == 6 & finish_flag == 1
		estadd scalar b_mean = r(mean)
		qui sum e_domain1 if treatment == 6 & e(sample)
		estadd scalar post_mean = r(mean)
			
		estimates store entropy_1

local iter = 1
foreach var in d1_1 d1_3 d1_7 d1_5 {
local ++iter

pdslasso `var' ib6.treatment (i.strata i.wave phone_survey survey_date missing_b`var' b`var' $cat_list $lik_list $con_list) [pweight=entropy_weight], partial(i.strata i.wave phone_survey survey_date missing_b`var' b`var') post(pds) robust cluster(ent_id) lopt(prestd)
		
		test 4.treatment = 5.treatment
		estadd scalar p1 = r(p)
		test 3.treatment = 5.treatment
		estadd scalar p2 = r(p)
		test 2.treatment = 5.treatment
		estadd scalar p3 = r(p)
		test 2.treatment = 4.treatment
		estadd scalar p4 = r(p)
		test 1.treatment = 2.treatment
		estadd scalar p5 = r(p)
		
		qui sum b`var' if treatment == 6 & finish_flag == 1
		estadd scalar b_mean = r(mean)
		qui sum `var' if treatment == 6 & e(sample)
		estadd scalar post_mean = r(mean)			
			
		estimates store entropy_`iter'
	}
	
pdslasso oyoh_yes ib6.treatment (i.strata $cat_list $lik_list $con_list) [pweight=entropy_weight], partial(i.strata) post(pds) robust cluster(ent_id) lopt(prestd)
		test 4.treatment = 5.treatment
		estadd scalar p1 = r(p)
		test 3.treatment = 5.treatment
		estadd scalar p2 = r(p)
		test 2.treatment = 5.treatment
		estadd scalar p3 = r(p)
		test 2.treatment = 4.treatment
		estadd scalar p4 = r(p)
		test 1.treatment = 2.treatment
		estadd scalar p5 = r(p)
		
		qui sum oyoh_yes if treatment == 6 & e(sample)
		estadd scalar post_mean = r(mean)
		estadd scalar b_mean = .
			
		estimates store entropy_6

esttab entropy_* using "$path/Output/Appendix_A/entropy.tex", label collabels(none) replace nolines nonumber keep(1.treatment 2.treatment 3.treatment 4.treatment 5.treatment) cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N b_mean post_mean p1 p2 p3 p4 p5, fmt(%9.0fc %12.2fc %12.2fc %9.2f %9.2f %9.2f %9.2f %9.2f) labels("Observations" "Control Mean: Baseline" "Control Mean: Follow-Ups" "Labeled Grant = Info Only" "Labeled Grant = Grant Only" "Labeled Grant = R-Mentee" "R-Mentee = Info Only" "R-Mentee = U-Mentee")) coeflabels(1.treatment "Mentored by Ugandan" 2.treatment "Mentored by Refugee" 3.treatment "Grant Only" 4.treatment "Information Only" 5.treatment "Info. + Labeled Grant") order(5.treatment 4.treatment 3.treatment 2.treatment 1.treatment) substitute(\_ _ \$ $) ///
$stars_setup ///
prehead("\begin{table}[h]	\centering	\footnotesize	\caption{Support for Refugee Integration (Weighted to Match Population Average Age, Education, and Policy Support)} \label{tab:entropy}	\begin{tabular}{l*{6}{>{\centering\arraybackslash}p{1.6cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-7}") ///
prefoot("& & & & & & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{7}{p{\linewidth}}{\footnotesize An observation is a surveyed respondent per post-baseline survey round in Uganda. Results estimated through ANCOVA regression with baseline controls selected through double lasso. All regressions use entropy balancing \citep{Hainmueller_2012} to match three moments to means in the Ugandan adult population: the share aged 18--30, the share completing secondary school, and the share that supports allowing refugees to work \citep{IRC_attitudes,UBOS2021}. Standard errors clustered at the enterprise level in parentheses; two-sided $ p $-values in brackets. $stars_note}  \end{tabular} \\ \end{table}%")

estimates drop entropy_*


*------------------------------------------------------------------------------------	
* Table A6 -- Full Phone Campaign Outcomes
*------------------------------------------------------------------------------------

label var oyoh_answered_call "\shortstack{Answered\\Call$^+$}"
label var oyoh_yes "\shortstack{Supported\\Phone\\Campaign$^+$}"
label var oyoh_no "\shortstack{Opposed\\Phone\\Campaign$^+$}"
 
local num = 0
foreach var in oyoh_yes oyoh_no oyoh_answered_call {
local ++num

pdslasso `var' ib6.treatment (i.strata $cat_list $lik_list $con_list), partial(i.strata) post(pds) robust lopt(prestd)
		test 4.treatment = 5.treatment
		estadd scalar p1 = r(p)
		test 3.treatment = 5.treatment
		estadd scalar p2 = r(p)
		test 2.treatment = 5.treatment
		estadd scalar p3 = r(p)
		test 2.treatment = 4.treatment
		estadd scalar p4 = r(p)
		test 1.treatment = 2.treatment
		estadd scalar p5 = r(p)
		
		qui sum `var' if treatment == 6 & e(sample)
		estadd scalar post_mean = r(mean)
					
estimates store oyoh_`num'
}

esttab oyoh_1 oyoh_2 oyoh_3  using "$path/Output/Appendix_A/oyoh_full.tex", label collabels(none) replace nolines nonumber keep(5.treatment 4.treatment 3.treatment 2.treatment 1.treatment) cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N post_mean p1 p2 p3 p4 p5, fmt(%9.0fc %12.2fc %9.2f %9.2f %9.2f %9.2f %9.2f) labels("Observations" "Control Mean: Follow-Ups" "Labeled Grant = Info Only" "Labeled Grant = Grant Only" "Labeled Grant = R-Mentee" "R-Mentee = Info Only" "R-Mentee = U-Mentee")) coeflabels(1.treatment "Mentored by Ugandan" 2.treatment "Mentored by Refugee" 3.treatment "Grant Only" 4.treatment "Information Only" 5.treatment "Info. + Labeled Grant") order(5.treatment 4.treatment 3.treatment 2.treatment 1.treatment) substitute(\_ _ \$ $) ///
$stars_setup ///
prehead("\begin{table}[h]	\centering	\footnotesize	\caption{Full Set of Phone Campaign Outcomes} \label{tab:oyoh_full}	\begin{tabular}{l*{3}{>{\centering\arraybackslash}p{2.5cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-4}") ///
prefoot("& & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{4}{p{0.8\linewidth}}{\footnotesize The sample is the experimental sample in Uganda. Results estimated through OLS regression with baseline controls selected through double lasso. Robust standard errors in parentheses. Brackets and the last five rows display two-sided $ p $-values. Outcomes that were not pre-specified are denoted with $^+$. $stars_note}  \end{tabular} \\ \end{table}%")

estimates drop oyoh_1 oyoh_2 oyoh_3


*------------------------------------------------------------------------------------
* Table A9 -- Business Outcomes and Household Welfare
*------------------------------------------------------------------------------------

*levels of profit and capital
foreach var in d2_1 d9_2{
	pdslasso `var'_stat ib6.treatment (i.strata i.wave phone_survey survey_date missing_b`var' b`var'_stat $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date missing_b`var' b`var'_stat) post(pds) robust cluster(ent_id) lopt(prestd)
	
		test 4.treatment = 5.treatment
		estadd scalar p1 = r(p)
		test 3.treatment = 5.treatment
		estadd scalar p2 = r(p)
		test 2.treatment = 5.treatment
		estadd scalar p3 = r(p)
		test 2.treatment = 4.treatment
		estadd scalar p4 = r(p)
		test 1.treatment = 2.treatment
		estadd scalar p5 = r(p)
		
		qui sum `var'_stat if treatment == 6 & e(sample)
		estadd scalar post_mean = r(mean)
		qui sum b`var'_stat if treatment == 6 & finish_flag == 1
		estadd scalar b_mean = r(mean)
		
		forvalues i=1/5{
			test `i'.treatment = 0
			local p`i' = r(p)
		}
		mat q = (`p1', `p2', `p3', `p4', `p5')
		mat colnames q = 1.treatment 2.treatment 3.treatment 4.treatment 5.treatment	
		estadd matrix q
			
estimates store est_`var'
}

label var d2_1_stat "\shortstack{Business\\Profits\\(USD/Month)}"
label var d9_2_stat "\shortstack{Business\\Capital\\(USD)}"

esttab sw_domain10 est_d2_1 est_d9_2 sw_domain9 using "$path/Output/Appendix_A/welfare.tex", label collabels(none) replace nolines nonumber keep(1.treatment 2.treatment 3.treatment 4.treatment 5.treatment) cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N b_mean post_mean p1 p2 p3 p4 p5, fmt(%9.0fc %12.2fc %12.2fc %9.2f %9.2f %9.2f %9.2f %9.2f) labels("Observations" "Control Mean: Baseline" "Control Mean: Follow-Ups" "Labeled Grant = Info Only" "Labeled Grant = Grant Only" "Labeled Grant = R-Mentee" "R-Mentee = Info Only" "R-Mentee = U-Mentee")) coeflabels(1.treatment "Mentored by Ugandan" 2.treatment "Mentored by Refugee" 3.treatment "Grant Only" 4.treatment "Information Only" 5.treatment "Info. + Labeled Grant") order(5.treatment 4.treatment 3.treatment 2.treatment 1.treatment) substitute(\_ _ \$ $) ///
$stars_setup ///
prehead("\begin{table}[t]	\centering	\footnotesize	\caption{Business Outcomes and Household Welfare} \label{tab:welfare}	\begin{tabular}{l*{4}{>{\centering\arraybackslash}p{2cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-5}") ///
prefoot("& & & & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{5}{p{0.84\linewidth}}{\footnotesize An observation is a surveyed respondent per post-baseline survey round in Uganda. Results estimated through ANCOVA regression with baseline controls selected through double lasso. Standard errors clustered at the enterprise level in parentheses. Brackets and the last five rows display two-sided $ p $-values. $stars_note}  \end{tabular} \\ \end{table}%")


*------------------------------------------------------------------------------------
* Table A10 -- Recall of Treatments
*------------------------------------------------------------------------------------

gen b_discusses_refugees = b_l5 <=3
gen discusses_refugees = e2_l5 <=3 if e2_l5 != .
	replace discusses_refugees = e_l5 <=3 if e_l5 != .
la var discusses_refugees "\shortstack{Discussed\\Refugees$^+$}"
	
pdslasso discusses_refugees ib6.treatment (i.strata i.wave phone_survey survey_date b_discusses_refugees $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_discusses_refugees) post(pds) robust cluster(ent_id) lopt(prestd)
	
		test 4.treatment = 5.treatment
		estadd scalar p1 = r(p)
		test 3.treatment = 5.treatment
		estadd scalar p2 = r(p)
		test 2.treatment = 5.treatment
		estadd scalar p3 = r(p)
		test 2.treatment = 4.treatment
		estadd scalar p4 = r(p)
		test 1.treatment = 2.treatment
		estadd scalar p5 = r(p)
		
		qui sum discusses_refugees if treatment == 6 & e(sample)
		estadd scalar post_mean = r(mean)
		qui sum b_discusses_refugees if treatment == 6 & finish_flag == 1
		estadd scalar b_mean = r(mean)
		
		forvalues i=1/5{
			test `i'.treatment = 0
			local p`i' = r(p)
		}
		mat q = (`p1', `p2', `p3', `p4', `p5')
		mat colnames q = 1.treatment 2.treatment 3.treatment 4.treatment 5.treatment	
		estadd matrix q
			
estimates store est_disc


esttab domain19_1 domain19_2 domain19_3 est_disc using "$path/Output/Appendix_A/treatment_recall.tex", label $nostar collabels(none) replace nolines nonumber keep(1.treatment 2.treatment 3.treatment 4.treatment 5.treatment) cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N b_mean post_mean p1 p2 p3 p4 p5, fmt(%9.0fc %12.2fc %12.2fc %9.2f %9.2f %9.2f %9.2f %9.2f) labels("Observations" "Control Mean: Baseline" "Control Mean: Follow-Ups" "Labeled Grant = Info Only" "Labeled Grant = Grant Only" "Labeled Grant = R-Mentee" "R-Mentee = Info Only" "R-Mentee = U-Mentee")) coeflabels(1.treatment "Mentored by Ugandan" 2.treatment "Mentored by Refugee" 3.treatment "Grant Only" 4.treatment "Information Only" 5.treatment "Info. + Labeled Grant") order(5.treatment 4.treatment 3.treatment 2.treatment 1.treatment) substitute(\_ _ \$ $) ///
$stars_setup ///
prehead("\begin{table}[h]	\centering	\footnotesize	\caption{Program Associations, Recall, and Discussions of Refugees} \label{tab:treatment_recall}	\begin{tabular}{l*{4}{>{\centering\arraybackslash}p{2cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-5}") ///
prefoot("& & & & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{5}{p{0.84\linewidth}}{\footnotesize Reports of support---and associations with YARID and data firm---are measured without prompting in a question about aid received from NGOs. An observation is a surveyed respondent per post-baseline survey round in Uganda. Results estimated through ANCOVA regression with baseline controls selected through double lasso. Standard errors clustered at the enterprise level in parentheses. Brackets and the last five rows display two-sided $ p $-values. Outcomes that were not pre-specified are denoted with $^+$. $stars_note}  \end{tabular} \\ \end{table}%")



*------------------------------------------------------------------------------------
* Table A11 -- Fairness of Aid Distribution
*------------------------------------------------------------------------------------

foreach q in aid1 aid1b aid3 aid5 aid7 aid9 aid10 aid11 aid12 aid13 aid14 {
	gen `q' = .
	count if e2_`q' > 3 & e2_`q' != .
	local num_high = r(N)
	count if e2_`q' < 3
	local num_low = r(N)
	replace `q' = (e2_`q' < 3)	if e2_`q' != . & `num_low' >= `num_high' 	
	replace `q' = (e2_`q' <= 3)	if e2_`q' != . & `num_low' < `num_high' 	
}

gen aid2_ref = e2_aid2_1 == 1 if e2_aid2_1 != .
gen aid8_refmore = e2_aid8 == 1 if e2_aid8 != .

foreach var in aid1 aid7 aid8_refmore aid13 aid11 aid12 {

pdslasso `var' ib6.treatment (i.strata phone_survey survey_date $cat_list $lik_list $con_list), partial(i.strata phone_survey survey_date) post(pds) robust cluster(ent_id) lopt(prestd)
	
		test 4.treatment = 5.treatment
		estadd scalar p1 = r(p)
		test 3.treatment = 5.treatment
		estadd scalar p2 = r(p)
		test 2.treatment = 5.treatment
		estadd scalar p3 = r(p)
		test 2.treatment = 4.treatment
		estadd scalar p4 = r(p)
		test 1.treatment = 2.treatment
		estadd scalar p5 = r(p)
		
		qui sum `var' if treatment == 6 & e(sample)
		estadd scalar post_mean = r(mean)
		estadd scalar b_mean = .
		
		forvalues i=1/5{
			test `i'.treatment = 0
			local p`i' = r(p)
		}
		mat q = (`p1', `p2', `p3', `p4', `p5')
		mat colnames q = 1.treatment 2.treatment 3.treatment 4.treatment 5.treatment	
		estadd matrix q
			
estimates store est_`var'
}

label var aid1 "\shortstack{Int'l Aid Is\\Distributed\\Fairly$^+$}"
label var aid7 "\shortstack{Refugees\\Get Too\\Much Aid$^+$}"
label var aid8_refmore "\shortstack{Refugees\\Get More\\Aid$^+$}"
label var aid13 "\shortstack{Local Aid\\Orgs Care\\About Me$^+$}"
label var aid11 "\shortstack{Int'l Aid\\Orgs Care\\About Me$^+$}"
label var aid12 "\shortstack{Int'l Aid\\Orgs Are\\Trustworthy$^+$}"

esttab est_aid1 est_aid7 est_aid8_refmore est_aid13 est_aid11 est_aid12 using "$path/Output/Appendix_A/aid_fairness.tex", label $nostar collabels(none) replace nolines nonumber substitute(\_ _ \$ $) keep(1.treatment 2.treatment 3.treatment 4.treatment 5.treatment) cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N b_mean post_mean p1 p2 p3 p4 p5, fmt(%9.0fc %12.2fc %12.2fc %9.2f %9.2f %9.2f %9.2f %9.2f) labels("Observations" "Control Mean: Baseline" "Control Mean: Follow-Ups" "Labeled Grant = Info Only" "Labeled Grant = Grant Only" "Labeled Grant = R-Mentee" "R-Mentee = Info Only" "R-Mentee = U-Mentee")) coeflabels(1.treatment "Mentored by Ugandan" 2.treatment "Mentored by Refugee" 3.treatment "Grant Only" 4.treatment "Information Only" 5.treatment "Info. + Labeled Grant") order(5.treatment 4.treatment 3.treatment 2.treatment 1.treatment) ///
$stars_setup ///
prehead("\begin{table}[H]	\centering	\footnotesize	\caption{Perceived Fairness of Aid Distribution} \label{tab:aid_fairness}	\begin{tabular}{l*{6}{>{\centering\arraybackslash}p{2cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-7}") ///
prefoot("& & & & & & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{7}{p{0.85\linewidth}}{\footnotesize An observation is a surveyed respondent in the 26-month survey round in Uganda. Results estimated through ANCOVA regression with baseline controls selected through double lasso. Standard errors clustered at the enterprise level in parentheses. Brackets and the last five rows display two-sided $ p $-values. Outcomes that were not pre-specified are denoted with $^+$. $stars_note}  \end{tabular} \\ \end{table}%")


*------------------------------------------------------------------------------------
* Table A12 -- Heterogeneity in Public Goods Usage
*------------------------------------------------------------------------------------

gen e2_uses_hospital = e2_d21 > 0 if e2_d21 != .
gen e2_child_ref_school = h_e20

bys ent_id: egen uses_hospital = max(e2_uses_hospital)
bys ent_id: egen child_ref_school = max(e2_child_ref_school)

gen either_publicgood = max(uses_hospital,child_ref_school)

local num = 0
eststo clear
foreach var in uses_hospital child_ref_school either_publicgood {
	local ++num
	gen X = `var' // to produce table
		
	*areg e_domain1 ib6.treatment##X phone_survey survey_date b_domain1 con_b_age cat_b_p12*, a(strata) robust
	pdslasso e_domain1 ib6.treatment##X (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
	
	eststo h_publicservice_`num'	
	drop X		
}	

esttab h_publicservice_* using "$path/Output/Appendix_A/hetero_publicgoods.tex", label $nostar collabels(none) replace nolines nonumber substitute(\_ _ \$ $) ///
cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N, fmt(%9.0fc) labels("Observations")) ///
keep(1.treatment 2.treatment 3.treatment 4.treatment 5.treatment 1.X  1.treatment#1.X 2.treatment#1.X 3.treatment#1.X 4.treatment#1.X 5.treatment#1.X) ///
order(5.treatment#1.X 5.treatment 4.treatment#1.X 4.treatment 3.treatment#1.X 3.treatment 2.treatment#1.X 2.treatment 1.treatment#1.X 1.treatment 1.X) ///
coeflabels(1.treatment "Mentored by Ugandan" 2.treatment "Mentored by Refugee" 3.treatment "Grant Only" 4.treatment "Information Only" 5.treatment "Info. + Labeled Grant" 1.X "\textit{X}" 1.treatment#1.X "Mentored by Ugandan $\times$ \textit{X}" 2.treatment#1.X "Mentored by Refugee $\times$ \textit{X}" 3.treatment#1.X "Grant Only $\times$ \textit{X}" 4.treatment#1.X "Information Only $\times$ \textit{X}" 5.treatment#1.X "Info. + Labeled Grant $\times$ \textit{X}") ///
mtitle("\shortstack{Uses\\Hospitals}" "\shortstack{Children Go\\to School With\\Foreigners}" "\shortstack{Uses Hospitals\\Or Schools}") ///
$stars_setup ///
prehead("\begin{table}[H]	\centering	\footnotesize \caption{Heterogeneity in Impacts on Integration Policies Index (Public Good Usage)} \label{tab:hetero_publicgoods}	\begin{tabular}{l*{3}{>{\centering\arraybackslash}p{3cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-4}") ///
prefoot("& & &  \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{4}{p{0.85\linewidth}}{\footnotesize The dependent variable for each column is the integration policies summary index. Each column title lists the dimension of heterogeneity (\textit{X})---which in this table is measured AFTER treatment---that is analyzed in the regression. An observation is a surveyed respondent per post-baseline survey round in Uganda. Results estimated through ANCOVA regression with baseline controls selected through double lasso. Standard errors clustered at the enterprise level in parentheses; two-sided $ p $-values in brackets. $stars_note}  \end{tabular} \\ \end{table}%")

estimates drop h_publicservice_*


*------------------------------------------------------------------------------------
* Table A13 -- Baseline Policy Attitude Drivers
*------------------------------------------------------------------------------------

label var bd1_1 "\shortstack{Supports\\Refugee\\Hosting}"
label var bd1_3 "\shortstack{Supports\\More\\Refugees}"
label var bd1_7 "\shortstack{Supports\\Right\\to Work}"
label var bd1_5 "\shortstack{Supports\\Freedom of\\Movement}"
label var b_domain1 "\shortstack{Integration\\Policies\\Index}"

local lasso_vars b_domain2 b_domain3 b_domain4 b_domain52 b_domain6 b_domain7 b_domain8 b_domain9 b_domain10

local iter = 0
foreach var in b_domain1 bd1_1 bd1_3 bd1_7 bd1_5 {
	local ++iter
eststo bcorr_`iter': rlasso `var' `lasso_vars' if wave_flag == 1, ols robust prestd

qui sum `var' if wave_flag == 1
estadd scalar b_mean = r(mean)
}

esttab bcorr_* using "$path/Output/Appendix_A/baseline_corr.tex", label $nostar collabels(none) replace nolines nonumber cells(b(fmt(%9.2f))) stats(N b_mean, fmt(%9.0fc %9.2f) labels("Observations" "Outcome Mean"))  substitute(\_ _ \$ $) ///
order(b_domain4 b_domain6 b_domain3 b_domain2 b_domain10) keep(b_domain4 b_domain6 b_domain3 b_domain2 b_domain10) coeflabels(b_domain4 "Economic Beliefs About Refugees" b_domain6 "Cultural Views About Refugees" b_domain3 "Knowledge of Hosting Policy" b_domain2 "Business Profit" b_domain10 "Household Well-Being") ///
prehead("\begin{table}[h]	\centering	\footnotesize	\caption{Baseline Correlates of Support for Refugee Integration} \label{tab:baseline_corr}	\begin{tabular}{l*{5}{>{\centering\arraybackslash}p{1.9cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-6}") ///
prefoot("& & & & & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{6}{p{\linewidth}}{\footnotesize Each column shows post-estimation OLS coefficients from a regression of a baseline policy outcome on the set of other primary and attitudinal domain summary indices among the experimental sample in Uganda. All domain summary indices normalized to mean 0, standard deviation 1.}  \end{tabular} \\ \end{table}%")

estimates drop bcorr_*

*------------------------------------------------------------------------------------
* Table A14 -- Heterogeneity Horserace Regression
*------------------------------------------------------------------------------------

qui sum b_domain10 if wave_flag == 1, d
cap gen h_domain10 = b_domain10 >= `r(p50)' 	if b_domain10 != .

eststo horse_1: pdslasso e_domain1 ib6.treatment##h_domain4 ib6.treatment##h_domain6 (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
eststo horse_2: pdslasso e_domain1 ib6.treatment##h_domain4 ib6.treatment##h_domain10 (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
eststo horse_3: pdslasso e_domain1 ib6.treatment##h_domain6 ib6.treatment##h_domain10 (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
eststo horse_4: pdslasso e_domain1 ib6.treatment##h_domain4 ib6.treatment##h_domain6 ib6.treatment##h_domain10 (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list), partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)


esttab horse_* using "$path/Output/Appendix_A/horserace.tex", label $nostar collabels(none) replace nolines nonumber substitute(\_ _ \$ $) ///
cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N, fmt(%9.0fc) labels("Observations")) ///
keep(3.treatment 4.treatment 5.treatment 3.treatment#1.h_domain4 4.treatment#1.h_domain4 5.treatment#1.h_domain4 3.treatment#1.h_domain6 4.treatment#1.h_domain6 5.treatment#1.h_domain6 3.treatment#1.h_domain10 4.treatment#1.h_domain10 5.treatment#1.h_domain10) ///
order(5.treatment#1.h_domain4 5.treatment#1.h_domain6 5.treatment#1.h_domain10 5.treatment 4.treatment#1.h_domain4 4.treatment#1.h_domain6 4.treatment#1.h_domain10 4.treatment 3.treatment#1.h_domain4 3.treatment#1.h_domain6 3.treatment#1.h_domain10 3.treatment 2.treatment#1.h_domain4 2.treatment#1.h_domain6 2.treatment#1.h_domain10 1.treatment#1.h_domain4 1.treatment#1.h_domain6 1.treatment#1.h_domain10) ///
coeflabels(3.treatment "Grant Only" 4.treatment "Information Only" 5.treatment "Info. + Labeled Grant" 3.treatment#1.h_domain4 "Grant Only $\times$ Pos. Economic" 4.treatment#1.h_domain4 "Information Only $\times$ Pos. Economic" 5.treatment#1.h_domain4 "Info. + Labeled Grant $\times$ Pos. Economic" 3.treatment#1.h_domain6 "Grant Only $\times$ Pos. Cultural" 4.treatment#1.h_domain6 "Information Only $\times$ Pos. Cultural" 5.treatment#1.h_domain6 "Info. + Labeled Grant $\times$ Pos. Cultural" 3.treatment#1.h_domain10 "Grant Only $\times$ High Well-Being" 4.treatment#1.h_domain10 "Information Only $\times$ High Well-Being" 5.treatment#1.h_domain10 "Info. + Labeled Grant $\times$ High Well-Being") ///
$stars_setup ///
prehead("\begin{table}[h]	\centering	\footnotesize	\caption{Expanded Treatment Effect Heterogeneity} \label{tab:horserace}	\begin{tabular}{l*{4}{>{\centering\arraybackslash}p{2.3cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-5}") ///
prefoot("& & & &  \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{5}{p{\linewidth}}{\footnotesize The dependent variable for each column is the integration policies summary index. \textit{Pos. Economic} indicates respondents with above-median beliefs about the economic impact of refugees at baseline. \textit{Pos. Cultural} indicates respondents with above-median cultural attitudes toward refugees at baseline. \textit{High Well-Being} indicates respondents with an above-median household well-being measure at baseline. All heterogeneity variables measured using domain summary indices. An observation is a surveyed respondent per post-baseline survey round in Uganda. Results estimated through ANCOVA regression with baseline controls selected through double lasso and include controls and interactions for both mentorship treatment groups (not shown). Standard errors clustered at the enterprise level in parentheses; two-sided $ p $-values in brackets. $stars_note}  \end{tabular} \\ \end{table}%")

estimates drop horse_*


*------------------------------------------------------------------------------------
* Table A15 -- Intensity of Mentorship Contact, and Timing by Survey Round
*------------------------------------------------------------------------------------

eststo intense_1: pdslasso e_domain1_fixed ib6.treatment_int (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list) if wave == 1, partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
	test 1.treatment_int = 2.treatment_int
	estadd scalar p_standard = r(p)
	test 7.treatment_int = 8.treatment_int
	estadd scalar p_intense = r(p)

eststo intense_2: pdslasso e_domain4_fixed ib6.treatment_int (i.strata i.wave phone_survey survey_date b_domain4 $cat_list $lik_list $con_list) if wave == 1, partial(i.strata i.wave phone_survey survey_date b_domain4) post(pds)  robust cluster(ent_id) lopt(prestd)
	test 1.treatment_int = 2.treatment_int
	estadd scalar p_standard = r(p)
	test 7.treatment_int = 8.treatment_int
	estadd scalar p_intense = r(p)

eststo intense_3: pdslasso e_domain6_fixed ib6.treatment_int (i.strata i.wave phone_survey survey_date b_domain6 $cat_list $lik_list $con_list) if wave == 1, partial(i.strata i.wave phone_survey survey_date b_domain6) post(pds) robust cluster(ent_id) lopt(prestd)
	test 1.treatment_int = 2.treatment_int
	estadd scalar p_standard = r(p)
	test 7.treatment_int = 8.treatment_int
	estadd scalar p_intense = r(p)
	
	local early_social = _b[5.treatment]

eststo intense_4: pdslasso e_domain1_fixed ib6.treatment_int (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list) if wave == 2, partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
	test 1.treatment_int = 2.treatment_int
	estadd scalar p_standard = r(p)
	test 7.treatment_int = 8.treatment_int
	estadd scalar p_intense = r(p)

eststo intense_5: pdslasso e_domain4_fixed ib6.treatment_int (i.strata i.wave phone_survey survey_date b_domain4 $cat_list $lik_list $con_list) if wave == 2, partial(i.strata i.wave phone_survey survey_date b_domain4) post(pds) robust cluster(ent_id) lopt(prestd)
	test 1.treatment_int = 2.treatment_int
	estadd scalar p_standard = r(p)
	test 7.treatment_int = 8.treatment_int
	estadd scalar p_intense = r(p)

eststo intense_6: pdslasso e_domain6_fixed ib6.treatment_int (i.strata i.wave phone_survey survey_date b_domain6 $cat_list $lik_list $con_list) if wave == 2, partial(i.strata i.wave phone_survey survey_date b_domain6) post(pds) robust cluster(ent_id) lopt(prestd)
	test 1.treatment_int = 2.treatment_int
	estadd scalar p_standard = r(p)
	test 7.treatment_int = 8.treatment_int
	estadd scalar p_intense = r(p)
	
	test 5.treatment = `early_social'
	
eststo intense_7: pdslasso e_domain1_fixed ib6.treatment_int (i.strata i.wave phone_survey survey_date b_domain1 $cat_list $lik_list $con_list) if wave == 4, partial(i.strata i.wave phone_survey survey_date b_domain1) post(pds) robust cluster(ent_id) lopt(prestd)
	test 1.treatment_int = 2.treatment_int
	estadd scalar p_standard = r(p)
	test 7.treatment_int = 8.treatment_int
	estadd scalar p_intense = r(p)

eststo intense_8: pdslasso e_domain4_fixed ib6.treatment_int (i.strata i.wave phone_survey survey_date b_domain4 $cat_list $lik_list $con_list) if wave == 4, partial(i.strata i.wave phone_survey survey_date b_domain4) post(pds) robust cluster(ent_id) lopt(prestd)
	test 1.treatment_int = 2.treatment_int
	estadd scalar p_standard = r(p)
	test 7.treatment_int = 8.treatment_int
	estadd scalar p_intense = r(p)
	
eststo intense_9: pdslasso e_domain6_fixed ib6.treatment_int (i.strata i.wave phone_survey survey_date b_domain6 $cat_list $lik_list $con_list) if wave == 4, partial(i.strata i.wave phone_survey survey_date b_domain6) post(pds) robust cluster(ent_id) lopt(prestd)
	test 1.treatment_int = 2.treatment_int
	estadd scalar p_standard = r(p)
	test 7.treatment_int = 8.treatment_int
	estadd scalar p_intense = r(p)
	
	test 5.treatment = `early_social'

esttab intense_* using "$path/Output/Appendix_A/impact_timing.tex", label collabels(none) replace nolines nonumber keep(1.treatment_int 2.treatment_int 3.treatment_int 4.treatment_int 5.treatment_int 7.treatment_int 8.treatment_int) cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N p_standard p_intense, fmt(%9.0fc %9.2f %9.2f) labels("Observations" "Refugee = Ugandan (Standard)" "Refugee = Ugandan (Intense)")) coeflabels(1.treatment_int "Standard Ugandan Mentorship" 2.treatment_int "Standard Refugee Mentorship" 3.treatment_int "Grant Only" 4.treatment_int "Information Only" 5.treatment_int "Info. + Labeled Grant"7.treatment_int "Intensive Ugandan Mentorship" 8.treatment_int "Intensive Refugee Mentorship") order(5.treatment_int 4.treatment_int 3.treatment_int 2.treatment_int 1.treatment_int 8.treatment_int 7.treatment_int) substitute(\_ _ \$ $) ///
$stars_setup ///
prehead("\begin{table}[H]	\centering	\footnotesize	\caption{More intensive refugee mentorship does not produce persistent impacts on policy views.} \label{tab:impact_timing}	\begin{tabular}{l*{9}{>{\centering\arraybackslash}p{1.3cm}}}\toprule \toprule &\multicolumn{3}{c}{9-Month Survey} &\multicolumn{3}{c}{16-Month Survey} & \multicolumn{3}{c}{26-Month Survey} \\ \cmidrule(l{2pt}r{2pt}){2-4} \cmidrule(l{2pt}r{2pt}){5-7} \cmidrule(l{2pt}r{2pt}){8-10}") ///
posthead("\cmidrule{2-10}") ///
prefoot("& & & & & & & & & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{10}{p{0.98\linewidth}}{\footnotesize An observation is a surveyed respondent per post-baseline survey round in Uganda. Results estimated through ANCOVA regression with baseline controls selected through double lasso. Each domain summary index is re-computed with a fixed set of components for comparability across survey rounds. Each set of 3 columns estimates impacts within a single post-intervention survey round. \textit{Intensive Mentorship} was offered to 100 business owners: these owners started their mentorship meetings earlier and so had more in-person and total meetings. \textit{Standard Mentorship} refers to those assigned to mentorship but not in the intensive group. Robust standard errors in parentheses; two-sided $ p $-values in brackets.}  \end{tabular} \\ \end{table}%")

estimates drop intense_*      




*------------------------------------------------------------------------------------
* Table C1 -- Child Labor 
*------------------------------------------------------------------------------------

// pool Cash + Info, R-mentee, and U-mentee with control

local num = 1
foreach var in d21_1 d21_2 {
	
	pdslasso `var' ib6.treatment_childlabor (i.strata phone_survey survey_date $cat_list $lik_list $con_list), partial(i.strata phone_survey survey_date) post(pds) robust cluster(ent_id) lopt(prestd)
	
	test 3.treatment_childlabor = 4.treatment_childlabor
	estadd scalar p = r(p)
	
	qui sum `var' if treatment_childlabor == 6 & wave == 4
	estadd scalar e2_mean = r(mean)		
	
	eststo domain21_`num'
	local ++num		
}
	
	* Domain Summary		
	pdslasso e_domain21 ib6.treatment_childlabor (i.strata phone_survey survey_date $cat_list $lik_list $con_list), partial(i.strata phone_survey survey_date) post(pds) robust cluster(ent_id) lopt(prestd)
		
	test 3.treatment_childlabor = 4.treatment_childlabor
	estadd scalar p = r(p)
	estadd scalar e2_mean = 0
		
	eststo sw_domain21
		
	esttab sw_domain21 domain21_1 domain21_2 using "$path/Output/Appendix_C/child_labor.tex", label $nostar collabels(none) replace nolines nonumber substitute(\_ _ \$ $) keep(3.treatment_childlabor 4.treatment_childlabor) cells(b(star fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N e2_mean p, fmt(%9.0fc %9.2f %9.2f) labels("Observations" "Control Mean" "Grant = Info")) coeflabels(3.treatment_childlabor "Grant Only" 4.treatment_childlabor "Information Only") order(3.treatment_childlabor 4.treatment_childlabor) ///
$stars_setup ///
prehead("\begin{table}[h]	\centering	\footnotesize	\caption{Impact of Child Labor Information Campaign} \label{tab:child_labor}	\begin{tabular}{l*{3}{>{\centering\arraybackslash}p{3cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-4}") ///
prefoot("& & & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{4}{p{0.8\linewidth}}{\footnotesize An observation is a surveyed respondent in the 26-month survey round in Uganda. Results estimated through OLS regression with baseline controls chosen through double lasso. Robust standard errors in parentheses; two-sided $ p $-values in brackets. $stars_note}  \end{tabular} \\ \end{table}%")
	
	estimates drop sw_domain21 domain21_1 domain21_2
	
	
	

*------------------------------------------------------------------------------------
* Table C2 -- Priming Analysis 
*------------------------------------------------------------------------------------

local num = 1
foreach var in prime_index d5a_1 d5a_2 d5b_1 d6_3 d4_4 {
	pdslasso `var' i.e_prime_treat (i.strata_prime ib6.treatment phone_survey survey_date b`var' missing_b`var' $cat_list $lik_list $con_list), partial(i.strata_prime ib6.treatment phone_survey survey_date b`var' missing_b`var') post(pds) robust cluster(ent_id) lopt(prestd)
	
	qui sum `var' if e_prime_treat == 0 & wave == 2
	estadd scalar e_mean = r(mean)	
	
	eststo prime_`num'
	
	local ++num
}
	esttab prime_* using "$path/Output/Appendix_C/prime.tex", label $nostar collabels(none) replace nolines nonumber substitute(\_ _ \$ $) keep(1.e_prime_treat) cells(b($stars_b fmt(%9.2f)) se(par fmt(%9.2f)) p(par([ ] ) fmt(%9.2f))) stats(N e_mean, fmt(%9.0fc %12.2f) labels("Observations" "Control Mean")) coeflabels(1.e_prime_treat "Primed on Aid Received$^+$") ///
$stars_setup ///
prehead("\begin{table}[h]	\centering	\footnotesize	\caption{Within-Survey Priming Experiment} \label{tab:prime}	\begin{tabular}{l*{6}{>{\centering\arraybackslash}p{1.4cm}}}\toprule \toprule ") ///
posthead("\cmidrule{2-7}") ///
prefoot("& & & & & & \\") ///
postfoot("\bottomrule \bottomrule \multicolumn{7}{p{\linewidth}}{\footnotesize An observation is a surveyed respondent in the 16-month survey in Uganda. Results estimated through OLS regression with baseline controls chosen through double lasso. Robust standard errors in parentheses; two-sided $ p $-values in brackets. Outcomes not pre-specified denoted with $^+$. $stars_note}  \end{tabular} \\ \end{table}%")
	
estimates drop prime_*





